

1 PHEASANT LANE

January 11, 1980

ITHACA, NY 14850

EFFICIENT CALCULATION OF THE FOURIER SERIES
OF RECTANGULAR FUNCTIONS

(607)-273-8030

The decomposition of a periodic function into sinusoidal components by the Fourier series (FS) is common practice. One form of the series that is commonly found is:

$$f(x) = a_0/2 + \sum_{n=1}^{\infty} [a_n \cos(nx) + b_n \sin(nx)] \quad (1)$$


While (1) tells us what the FS is, it does not tell us how to find the Fourier coefficients a_n and b_n . By far the most widely known method of obtaining the Fourier coefficients is through the integration formulas (see just about any book on engineering math or Fourier methods for the derivation):

$$a_n = (1/\pi) \int_{-\pi}^{\pi} f(x) \cos(nx) dx \quad n = 0, 1, 2, \dots \quad (2)$$

$$b_n = (1/\pi) \int_{-\pi}^{\pi} f(x) \sin(nx) dx \quad n = 1, 2, 3, \dots \quad (3)$$

A number of equivalent integration formulas are also common. Essentially these formulas reduce the determination of the coefficients to the problem of integrating the products $f(x)\sin(nx)$ and $f(x)\cos(nx)$. When $f(x)$ takes on only constant values over an interval, the integral is done over each interval separately, and we have only to integrate $\sin(nx)$ and $\cos(nx)$ since $f(x)$ is a constant over these intervals. With such "piecewise constant" functions, integration is reduced to bookkeeping of values of $\sin(nx)$ and $-\cos(nx)$ at the ends of the intervals - straightforward, but tedious for some functions. For "rectangular" functions, those which take on only one of two possible levels at any one time, things are theoretically very simple, but these "digital sequences" can be rather complicated in structure (segments of pseudo-random sequences, Walsh functions, complex pulse trains, etc.), and we may well wonder if there is a simpler formula in such cases. There is.

It should be understood that method used here can be generalized to any piecewise constant function (not just the 0 and 1 levels we will use), and through the use of jumps in the derivatives of $f(x)$, to a much wider class of function [see E. Kreyszig, Advanced Engineering Mathematics, John Wiley (1967)]. Thus the rectangular functions we will consider are a special simple case. To understand how basically simple the method is, consider the portion of a rectangular function shown at the right. It has jumps of +1 and -1 (denoted j_s) at points denoted x_s . The formula for a_n above (2) can be used for illustration.



$$\begin{aligned} a_n &= (1/\pi) \int_{-\pi}^{\pi} f(x) \cos(nx) dx \\ &= (1/n\pi) \left\{ \dots \sin(nx) \Big|_{x_s}^{x_{s+1}} + \sin(nx) \Big|_{x_{s+2}}^{x_{s+3}} + \dots \right\} \\ &= (1/n\pi) [\dots \sin(nx_{s+1}) - \sin(nx_s) + \sin(nx_{s+3}) - \sin(nx_{s+2}) + \dots] \\ &= (-1/n\pi) [\dots j_{s+1} \sin(nx_{s+1}) + j_s \sin(nx_s) + j_{s+3} \sin(nx_{s+3}) + j_{s+2} \sin(nx_{s+2}) \dots] \\ &= (-1/n\pi) \sum_{s=a}^b j_s \sin(nx_s) \end{aligned}$$

Where a and b are the jumps at the beginning and end of a full cycle respectively. Note that b is not the jump a one cycle later, but the jump just before the repeat of a. Because we are using rectangular functions with jumps only +1 and -1, we can simplify the result and at the same time write the corresponding equation for b_n . Here we assume that j_1 is +1, that is, we start the function cycle on a positive jump. The equations become:

$$a_n = (1/n\pi) \sum_{s=1}^b (-1)^s \sin(nx_s) \quad (4)$$

$$b_n = (-1/n\pi) \sum_{s=1}^b (-1)^s \cos(nx_s) \quad (5)$$

Equations (4) and (5) thus reduce the calculation to the sum of a series with the only needed input to the series being the jump points x_s .

Equations (4) and (5) can be easily implemented on a programmable calculator of just about any type. Below we show a TI-59 PC-100A program of this type. To use the program, first enter the overall period of one cycle in convenient units, and press A. Next, enter the position of the first positive jump (in the same units) and press B. Enter the additional jump points, pressing B after each entry. You may enter up to 38 such jumps. After the final jump (which should be negative of course), enter 8888 and press B one last time. [8888 is just an arbitrary recognition number - any other value could be used if 8888 happens to be a jump point, which is unlikely.] To run the program, press C, and it will first print out the DC value of the function, followed by a sequence of four numbers for each harmonic. These four numbers are N, the order of the harmonic, AN (a_n), BN (b_n), and CN ($c_n = \sqrt{a_n^2 + b_n^2}$). When enough harmonics are printed out, press R/S to stop the program.

```

2nd Lbl A 2nd CMs STO 41 2nd RAD 1 STO 42 STO 43 0 STO 49 8888 x>t R/S
2nd Lbl B STO 2nd Ind 42 1 SUM 42 R/S
2nd Lbl C 2nd D' 2nd Lbl 2nd A' 1 STO 42 STO 45 0 STO 46 STO 47
  2nd Lbl 2nd C' RCL 2nd Ind 42 x=t D x 2 x 2nd pi x RCL 43 ÷ RCL 41 = STO 44
  RCL 44 2nd Sin x RCL 45 = SUM 46 RCL 44 2nd COS x
  RCL 45 = SUM 47 RCL 45 +/- STO 45 1 SUM 42 GTO 2nd C'
  2nd Lbl D RCL 46 ÷ RCL 43 ÷ 2nd pi = +/- STO 48 RCL 47 ÷ RCL 43 ÷ 2nd pi = STO 49
  RCL 48 x^2 + (RCL 49 x^2) = sqrt STO 50
31 2nd Op 04 RCL 43 2nd Op 06 1331 2nd Op 04 RCL 48 2nd Op 06
1431 2nd Op 04 RCL 49 2nd Op 06 1531 2nd Op 04 RCL 50 2nd Op 06
  2nd Adv 1 SUM 43 GTO 2nd A'
2nd Lbl 2nd D' 1 STO 42 2 STO 48
  2nd Lbl E RCL 2nd Ind 48 - RCL 2nd Ind 42 x=t 2nd E' SUM 49 2 SUM 42 SUM 48 GTO E
  2nd Lbl 2nd E' RCL 49 ÷ RCL 41 = +/- STO 49 1615 2nd Op 04 RCL 49 2nd Op 06
  2nd Adv INV SBR
  
```

We can illustrate the method on the square wave shown at the right. It has a jump of +1 at $x = 0$, and a jump of -1 at $x = \pi$. Using equation (5) for b_n , we have:

$$\begin{aligned} b_n &= (-1/n\pi) [(-1)^1 \cos(n \cdot 0) + (-1)^2 \cos(n \cdot \pi)] \\ &= (-1/n\pi) [-1 + \cos(n\pi)] \\ &= 2/n\pi \text{ for } n \text{ an odd number} \\ &= 0 \text{ for } n \text{ an even number} \end{aligned}$$

In the same way, we find that $a_n = 0$ in all cases. This is the familiar result of a square wave consisting of odd harmonics falling off as $1/n$.

